BOOTVAR

USER GUIDE
(BOOTVAR 3.1 - SAS VERSION)

STATISTICS CANADA
APRIL 2005

TABLE OF CONTENTS

1. Introduction 3

Changes from the Previous Version 3

2. Bootstrap Method 4

3. Description of steps for using BOOTVAR 5

Step 1: Creation of the Analysis File 5

Step 2: Variance Calculation Using BOOTVAR.SAS 6

4. Additional tips and options for using BOOTVAR.SAS 7

5. Alternatives to BOOTVAR when using the bootstrap method 8

Appendix A: Fact Sheets 9

Appendix B: Example 10

1. Introduction

This guide is for users of the SAS version of BOOTVAR, which was created to estimate variances using the bootstrap method. BOOTVAR allows for the estimation of variances for totals, ratios (including proportions), differences between ratios (or proportions), percentiles, linear regression models, logistic regression models and Chi-square tests of independence. BOOTVAR does not generate bootstrap weights, but uses those provided with the survey data.

Section 2 of this guide briefly explains the bootstrap resampling method used to estimate the variance. Section 3 describes in detail how BOOTVAR works and the preliminary steps that are required to run the program. Additional tips for using the program are provided in Section 4, including the optional modification of parameters. The fact that bootstrap variance estimation can be accomplished using other commercially available software is discussed in Section 5. Three appendices accompany the User Guide. Appendix A contains fact sheets presenting the details of each of the BOOTVAR macros. Appendix B contains a complete example (programs and results). Finally, the survey-specific parameters for a number of Statistics Canada Surveys using the bootstrap method, required for executing the programs, are provided in Appendix C. The latter is not included in the current document, but constitutes a separate document, distributed with BOOTVAR, or available on demand via the general contact information provided in the survey documentation for each survey.

Changes from the Previous Version:

Beginning with version 3.0, BOOTVAR is considered generic in that it attempts to support all Statistics Canada surveys that use the bootstrap method for variance estimation. The generic nature of the program requires only that the user specify a few additional (relative to earlier versions) parameters in section 1 of BOOTVAR. The values to specify for these parameters are presented in Appendix C.

Beginning with version 2.0, BOOTVAR consists of two programs. In version 3.1, these two programs are called BOOTVARE_V31.SAS and MACROE_V31.SAS, and are described in Section 3. In order to shorten the text of the User Guide, these two programs will be referred to by the abbreviations BOOTVAR.SAS and MACRO.SAS.

The modifications made in version 3.1 (with respect to version 3.0) are:

- Addition of a macro to estimate variances for percentiles.
- Addition of a macro to conduct Chi-square tests of independence.
- Addition of a parameter allowing for the use of the mean bootstrap method.
- Addition of sample sizes to the output statistics for totals, ratios and differences of ratios.
- Addition of t-statistics and the corresponding p-values to the output statistics for a linear regression.
- Addition of z-statistics and the corresponding p-values to the output statistics for a difference of ratios.
- Possibility of modifying the default significance level used in calculating confidence intervals.

- Possibility of using variables with negative values when estimating the variance of a total, a ratio or a difference of ratios.
- Possibility of calculating the design effect for a total or a ratio.

It is important to note that BOOTVAR version 3.1 for SAS has been tested and works with versions 6.12 and 8.2 of SAS. Appropriate results are not guaranteed when using the program with older or more recent versions of SAS.

2. Bootstrap Method

Many Statistics Canada surveys use complex sampling designs when selecting their samples. As variance estimation for these sampling schemes cannot be accomplished using simple formulae, we must use approximate methods to estimate variances. Resampling methods, and in particular the bootstrap method, figure among these. The bootstrap approach possesses many interesting properties and is the method recommended by many Statistics Canada surveys.

Briefly, the bootstrap method consists of drawing several sub-samples from the full sample. These sub-samples are the result of a simple random sample (SRS) with replacement of n-1 clusters, among the n clusters selected within each stratum. The number of sub-samples, B, varies from survey to survey according to individual needs and objectives of the survey. An adjusted weight, specific to each sub-sample (also called replicate), is assigned to each unit belonging to the bootstrap sub-sample. This adjusted weight is referred to as a bootstrap weight. To estimate the variance for a point estimate (a statistic calculated from the sampling weight), it is sufficient to calculate this same point estimate B times using the B bootstrap weights. The variability among the B estimates provides the variance estimate.

The bootstrap weights are produced and provided by the survey. BOOTVAR uses these weights to estimate variances, as well as other measures of variability, such as the standard error, confidence interval and coefficient of variation. These measures should be used to determine whether or not a point estimate should be published (please consult the survey-specific guidelines for publication of results), or to calculate test statistics.

In summary, here are the main steps carried out by BOOTVAR to calculate the variance of a given point estimate:

a) The point estimate (total, ratio, etc.) is calculated using the sampling weight included on the data file.
b) The same statistic is calculated using each of the B bootstrap weights on the bootstrap weights file. B bootstrap estimates (of the total, ratio, etc.) are thus obtained.
c) Finally, the variance (according to the formula used for a simple random sample) of the B bootstrap estimates is calculated. This variance corresponds to the estimated variance of the point estimate calculated in a).

There is a great deal of literature dealing with the bootstrap method and its use with data from surveys of complex design. For those who wish to further their knowledge of the bootstrap method, two such references are provided below. The first reference (Rao et Wu, 1988) is a more technical paper on the theory surrounding the used of the bootstrap, while the second (Yeo, Mantel et Liu, 1999) presents an example of the application of the bootstrap in the survey setting.

- Rao, J.N.K. and Wu, C.F.J. (1988). *Resampling Inference with Complex Survey Data*. Journal of the American Statistical Association. Vol. 83, No. 401, 231-241.
- Yeo, D., Mantel, H. and Liu, T-P. (1999). *Bootstrap Variance Estimation for the National Population Health Survey*. 1999 Proceedings of the Survey Research Methods Section, American Statistical Association, pp. 778-783.

3. Description of steps for using BOOTVAR

The BOOTVAR program is a set of macros, where each macro estimates variances for a particular statistic. The fact sheets in Appendix A describe the macros available in this version of BOOTVAR, and provide the necessary information for their use. It is essential to consult these fact sheets in order to acquaint oneself with the constraints and limitations of the macros.

Variance estimation is performed in ***two steps*** and involves the use of three SAS programs. The ***first step*** consists of creating a data file containing the variables required for the analysis (first program). The ***second step*** involves using BOOTVAR.SAS (and MACRO.SAS) to estimate the variances.

Step 1: Creation of the Analysis File

The user needs to create a SAS data file which will be used as the input file for the program that estimates the variance in step 2. The following tasks must be done in this step:

1. Reading of the input file
2. Creation of the variables required for the analysis

1 - <u>Reading of the input file</u>: The analysis file is created from the survey data file. Typically, the file is read using a file layout provided with the data. Appendix C provides the names of the data files associated with each survey that endorses BOOTVAR.

2 - <u>Creation of the variables required for the analysis</u>: Variables derived from the input variables should be created in this step. It may be necessary to create dichotomous variables (1 or 0) which identify records that have a particular characteristic – such variables will take a value of 1 for records that have the characteristic and a value of 0 otherwise. For example, when estimating totals, ratios and differences between ratios, these dichotomous variables will identify which records possess the characteristic of interest, in order to sum their weights to obtain the desired estimate at step 2. See the example in Appendix B for more details.

The analysis file must contain:

- The necessary variables for the analysis (derived variables including dichotomous variables, and input variables that do not need to be modified). To reduce the runtime of the program, DO NOT keep unnecessary variables.
- The unique identification variable(s) for units in the sample. See Appendix C to obtain the name of the unique identification variable(s).
- If needed, the breakdown variable(s), identifying the groups for which a separate analysis is desired (ex.: province, gender, etc.).
- If the analysis is to be carried out only for a certain subgroup (for example, a province or an age group), keep only the records that belong to this subgroup in order to reduce runtime.

<u>REMARKS:</u>

- At this step, it is recommended that point estimates be produced based on the sampling weight provided on the survey data file. As BOOTVAR also calculates the point estimate, it is then possible for the user to validate his/her work by ensuring that the point estimate produced by BOOTVAR does indeed correspond to that calculated initially. Differences between the two results will indicate that the parameters specified in BOOTVAR are not replicating the concept measured at the first step. Please note that in order to produce the initial point estimate, the user must be sure to keep the weight variable when creating the analysis file.

The user must create their own program to prepare the SAS data file containing the necessary variables for the analysis. An example of a program that creates this file is included in Appendix B (the program STEP1.SAS).

Step 2: Variance Calculation Using BOOTVAR.SAS

Once the new SAS data file is created in step 1, the next step consists of running the BOOTVAR.SAS program. Before running it, the desired parameters and analyses must be specified. This program calls the MACRO.SAS program. MACRO.SAS contains the program code of the various macros. *For standard use of the variance estimation program, no modification of the MACRO.SAS program by the user is necessary.* That being said, as BOOTVAR is distributed as open source code, it is possible for users experienced in SAS programming to modify the program code in order to satisfy needs not addressed by BOOTVAR. In such situations, it is recommended that the program MACRO.SAS be renamed in order to avoid confusion with the original version of the program.

An example of how to use the BOOTVAR.SAS program is included in Appendix B. The parts that are to be changed by the user are given in ***bold type.*** The rest of the program does not need to be changed. The program is divided into ***two sections***. The ***first section*** is for specifying the required parameters, and the ***second section*** is for listing the desired analyses.

Section 1:

In this section, the user must specify:

- The name of the directory where the analysis file created in step 1 is located; and the name of the directory to which the output file containing the results produced by BOOTVAR are to be saved
- The name of the data file (analysis file) created in step 1
- The name of the directory and file containing the bootstrap weights
- The name(s) of breakdown variable(s), if the analysis is to be performed separately for specific sub-groups (ex.: provinces, gender)
- The survey-specific parameters, i.e.: the name(s) of the unique identification variable(s) of the respondents, the name of the survey weight variable, the prefix used to name the bootstrap weight variables, the mean bootstrap parameter and the number of bootstrap weights to be used. Appendix C contains the necessary information to correctly assign the parameter values, by survey.
- The name of the directory where the program MACRO.SAS is located

Section 2:

The user specifies the desired analyses. Variance estimates can be obtained for:

- Totals
- Ratios (including proportions and means)
- Differences between ratios
- Percentiles
- Regression models (linear or logistic)
- Chi-square tests of independence

Please refer to Appendix A for a detailed description of the different types of analysis and for a description of the different results produced by the corresponding macro.

4. Additional tips and options for using BOOTVAR.SAS

- Modifying BOOTVAR while testing programs: The runtime for certain macros can be rather lengthy (particularly for regression and Chi-square tests). It is recommended to test one's program with a small number of bootstrap weights (say 10) before the final run. **It is imperative to use all bootstrap weights provided by a particularly survey when calculating final variance estimates.** The number of bootstrap weights can be modified in the first part of the BOOTVAR.SAS program, by changing the value assigned to the macro variable *B*.

- Modifying the significance level for confidence intervals: The significance level used by default in the output is 5% (alpha=0.05). To modify the level when calculating confidence intervals, one need only change the default value of alpha by adding the statement `%let alpha=desired_level;` at the very beginning of Section 2 of the BOOTVAR.SAS program (before the call to macro `%total`). For example, to produce 90% confidence intervals, the statement `%let alpha=0.10;` would be used.

- Calculating the design effect for a total or a ratio: It is possible to calculate design effects when estimating variances for totals and ratios. This has made possible made adding the statement `%let deff=1;` at the very beginning of Section 2 of the BOOTVAR.SAS program (before the call to macro `%total`). Note that the use of this option will slightly increase the runtime.

5. Alternatives to BOOTVAR when using the bootstrap method

BOOTVAR is not the only tool capable of estimating variances and carrying out hypothesis tests using bootstrap weights. While the use of bootstrap weights is not explicitly supported by commercially available software such as SUDAAN and WesVar, by taking advantage of similarities between a commonly used bootstrap technique and the method of Balanced Repeated Replication (BRR), these software can be used to produce bootstrap variance estimates (see Phillips, 2004). Additionally, any software that offers an analytic procedure or command that can produce weighted estimates of the parameters of interest and also has the flexibility of a programming language, may be used recursively to obtain bootstrap variance estimates.

Based upon this principle, programs similar to BOOTVAR have been written for different software. For example, the user-defined Stata command Bswreg, can be used to obtain bootstrap variance estimates for many of Stata's existing regression commands. The benefits of this program are presented in Piérard et al, 2004.

- Phillips, O. (2004) Using Bootstrap Weights with WesVar and SUDAAN. *The Research Data Centres Information and Technical Bulletin* 1(2): 6-15.
- Piérard, E., Buckley, N., Chowhan, J. (2004) Bootstrapping made easy: A Stata ADO file. *The Research Data Centres Information and Technical Bulletin* 1(1): 20-36.

Appendix A: Fact Sheets

This section is still in development. Please send an e-mail to owen.phillips@statcan.ca indicating that you would like the updated version of the documentation when it is available.

Appendix B: Example

This is a complete example showing how to use the BOOTVAR.SAS program. First, the analysis data file is created (step 1). Then, BOOTVAR.SAS is adapted to obtain the desired analysis. The results that are produced follow the programs.

Example:

This example uses the cycle 3 (1998) cross-sectional file of the National Population Health Survey, general component. The analyses are done separately for each province (breakdown variable) an only four provinces are considered. The objectives are:

1- Compute the number of diabetics, by gender
2- Compute the proportion of diabetics for males and females
3- Compare diabetes rate for men to that of women
4- Compute the 75^{th} percentile for age
5- Study the relationship between age, diabetes and gender (linear regression)
6- Study the relationship between diabetes, gender and age (logistic regression)
7- Study the dependence of the relationship between diabetes and gender (Chi-square test of independence)

The parameters required to run BOOTVAR are (from Appendix C)

National Population Health Survey (NPHS) - Household Component							
	Name of data file	**Name of bootstrap weights file**	**Identification variable(s) (ident)**	**Weight variable (fwgt)**	**Prefixe of the bootstrap weights (bsw)**	**# of weights (B)**	**Mean bootstrap (R)**
Cross-sectional - General component :							
Cycle 3	H35	B5H35	REALUKEY PERSONID	FWGT	BSW	500	1

Step 1:

```
*************************************************************
*                    STEP1.SAS                              *
*                                                           *
*        This program creates the SAS datafile              *
*        containing the necessary variables                 *
*        for the BOOTVARE_V31.SAS program                   *
*************************************************************;


LIBNAME in1 'c:\bootvar';


*** Creation of the SAS data file containing the variables and cases required
*** for the analysis. Note that this file should be as small as possible (containing
*** only necessary variables and cases) in order to reduce time and memory requirements
*** especially if regression type analysis are to be done.  ;


data in1.diabetes;                 /* file to be used with BOOTVARE_V31.SAS */
    %let datafid= "D:\Data\h35.txt";
    %include " D:\Layout\h35_i.sas ";


***  Creation of Dichotomous Variables                                  ***

*** (examples are presented below using National Population Health ***
***  Survey, cycle 3 variables)                                     ***;


   ***  Keep only 4 provinces;
         if prc8_cur in (10 24 35 59);

   /* diabetes */
          if ccc8_1j=1 then diab=1;
          else if ccc8_1j=2 then diab=0;
          else diab=.;

   /* sex */
          if dhc8_sex=1 then males=1;
          else if dhc8_sex=2 then males=0;
          else males=.;
          if dhc8_sex=2 then females=1;
          else if dhc8_sex=1 then females=0;
          else females=0;

   /* diabetes*sex */
          mdiab = diab * males;      /* male diabetics   */
          fdiab = diab * females;    /* female diabetics */

   /* age */
          if DHC8_AGE > 200 then DHC8_AGE=.;

   keep realukey personid wt58 prc8_cur diab mdiab fdiab males females DHC8_AGE;


          * It is recommended that only the necessary variables be kept    *
          * in order to reduce the runtime of BOOTVARE_V31.SAS.            *
          * IMPORTANT:  the identification variables and, if necessary,    *
          * the breakdown variable (ex: province, sex) must be kept. The   *
          * weight variable also must be kept if point estimates are       *
          * calculated at this step                                        *;

run;
```

Step 2 – BOOTVARE_V31.SAS Program:

…
…

```
/*****************************************************************************/
/***                              SECTION 1                              ***/
/*****************************************************************************/
/***                                                                     ***/
/***   This section lets the user specify the different parameters of    ***/
/***   interest (variable names, directory names, file names, etc.)      ***/
/***                                                                     ***/
/*****************************************************************************/


***********************************************************************
** SPECIFY THE NAME OF THE FOLLOWING 2 DIRECTORIES (directories only): **
***********************************************************************;

      libname in1 "c:\bootvar";    /* (ex: c:\data) */
      libname out "c:\bootvar";                   /* (ex: c:\output)*/


***************************************************************************************
** SPECIFY THE NAME OF THE ANALYSIS FILE (CREATED IN STEP 1) (without extension):  **
***************************************************************************************;

      %let Mfile = in1.diabetes;


***********************************************************************
** SPECIFY THE NAME OF THE FILE CONTAINING THE BOOTSTRAP WEIGHTS:      **
**        NB: Only run one of the two following series of commands     **
**        (comment the other one out, or erase it):                    **
***********************************************************************;

    *  EXECUTE THIS PART IF THE BOOTSTRAP WEIGHTS ARE IN SAS FORMAT (remove the " * ")
       ----------------------------------------------------------------------------;

        *  libname in2 "directory_name_containing_bootstrap_weights_file"; /* (ex: c:\bootstrp)*/
        *  %let bsamp=in2.SAS_file_name_containing_the_weights_(without extension) ;

    *  EXECUTE THIS PART IF THE BOOTSTRAP WEIGHTS ARE IN ASCII (.TXT) FORMAT (remove the " * ")
       -------------------------------------------------------------------------------------;

                data bootwt;
                     %let datafid="D:\bootstrp\DATA\cb5h35.txt";
                     %include "D:\bootstrp\Layout\cb5h35_i.sas";
                run;
                %let bsamp=bootwt;

************************************************************************************************
** SPECIFY, IF DESIRED, THE BREAKDOWN VARIABLE(S) (EG: PROVINCE, SEX, ETC...):                **
**     Write the name of the breakdown variable(s) below.                                     **
**                                                                                            **
**   - If the analysis includes all of the data in the file created in step 1, put a dot.     **
**     (%let classes =. )                                                                     **
**   - If more than one variable, leave a blank between each variable (%let classes=var1 var2) **
**   - DO NOT ERASE OR COMMENT OUT THIS COMMAND                                               **
************************************************************************************************;

      %let classes = prc8_cur  ;


*********************************************************************************************
** SPECIFY THE FOLLOWING INFORMATION (SPECIFIC TO THE SURVEY YOU ARE USING):               **
**    You must specify:                                                                    **
**           1- The unique identifier variable(s) (separated by a space)                   **
**           2- The final weight (variable included in the bootstrap weight file)          **
**           3- The prefix of the boostraps weight variables                               **
**           4- Parameter for the mean bootstrap (set to 1 for regular bootstrap)          **
**           5- The number of bootstrap weights to use (note: For testing, B must be >= 2. **
**              IT IS NECESSARY TO USE ALL THE BOOTSTRAP WEIGHTS WHEN PERFORMING THE FINAL **
**              ANALYSIS.)                                                                 **
**                                                                                         **
**           - Refer to AppendixC  to obtain this information                              **
*********************************************************************************************;

      %let ident = realukey personid;
      %let fwgt  = fwgt;
      %let bsw   = bsw;
```

```
%let R      = 1;
%let B      = 500 ;


*********************************************************************************
** SPECIFY THE DIRECTORY AND THE NAME OF THE FILE THAT CONTAINS THE MACROS     **
** (THE PROGRAM MACROE_V31.SAS IF NO MODIFICATIONS HAVE BEEN MADE BY THE USER) **
*********************************************************************************;

      %include " D:\Bootstrp\pgm\SAS\MACROF_V31.SAS ";



/******************************************************************************/
/***                              SECTION 2                                ***/
/******************************************************************************/
/***                                                                       ***/
/*** This section lets the user specify the different analyses of interest.***/
/***                                                                       ***/
/******************************************************************************/


/**********************************************************************************/
/*** Variance estimates are calculated using the SAS macros defined in the       ***/
/*** MACROE_V31.SAS program.  These SAS macros can be submitted in this program, ***/
/*** to suit the user's needs.                                                   ***/
/***                                                                             ***/
/*** To submit a macro, the statement must specify the macro name and parameters ***/
/*** to use.  In this program, the parameters indicate which variables will be   ***/
/*** used.                                                                       ***/
/***                                                                             ***/
/*** Each macro submission gives an estimate of the variance of only one         ***/
/*** parameter. If more than one parameter and variance need to be               ***/
/*** calculated, the macro then must be submitted several times.                 ***/
/***                                                                             ***/
/*** A statement for each possible type of analysis appears in the program. They ***/
/*** are commented out and the user only needs to run those that are desired.    ***/
/***                                                                             ***/
/***                                                                             ***/
/*** COMMENT ABOUT DIFFERENCES BETWEEN RATIOS:                                   ***/
/***  ----------------------------------------                                   ***/
/***                                                                             ***/
/*** The confidence interval is calculated for a single comparison of ratios.  If***/
/*** multiple comparisons are made, the method used to calculate the confidence  ***/
/*** interval must take this into account.  For this reason,  in the case of     ***/
/*** multiple comparisons, the Z value from the normal distribution used in the  ***/
/*** calculation of the confidence interval must be corrected in the diff_rat    ***/
/*** macro in the MACROE_V31.SAS program using, for example, the Bonferroni      ***/
/*** approach for multiple comparisons.                                          ***/
/***                                                                             ***/
/**********************************************************************************/


* TO OBTAIN VARIANCE ESTIMATES OF A TOTAL, RUN:
  ---------------------------------------------;

        %total(mdiab);
        %total(fdiab);


* TO OBTAIN VARIANCE ESTIMATES OF A RATIO, RUN:
  ---------------------------------------------;

        %ratio(mdiab ,hommes );
        %ratio(fdiab ,femmes );


* TO OBTAIN VARIANCE ESTIMATES OF A DIFFERENCE BETWEEN RATIOS, RUN:
  -----------------------------------------------------------------;
                *NOTE: see the comment at the beginning of section 2 ... ;

        %diff_rat(mdiab,males,fdiab,females);

                        *  where:  VAR1   : the numerator variable of the first ratio      *
                        *          VAR2   : the denominator variable of the first ratio    *
                        *          VAR3   : the numerator variable of the second ratio     *
                        *          VAR4   : the denominator variable of the second ratio   *;
```

```
* TO OBTAIN VARIANCE ESTIMATES OF A PERCENTILE p (p between 1 and 99), RUN:
  -----------------------------------------------------------------------;

        %prcntle(dhc8_age,75);

* TO OBTAIN VARIANCE ESTIMATES OF LINEAR REGRESSION PARAMETERS, RUN:
  -----------------------------------------------------------------;

        %regress(dhc8_age,diab females);

* TO OBTAIN VARIANCE ESTIMATES OF LOGISTIC REGRESSION PARAMETERS, RUN:
  -------------------------------------------------------------------;

        %logreg(diab,females dhc8_age);

* TO EXECUTE A CHI-SQUARE TEST OF INDEPENDENCE, RUN:
  -------------------------------------------------;

        %chi2(diab,females);

%output;  /*Displays the results on the screen.  Do not modify. */

* TO SAVE THE RESULTS IN A FILE, RUN: (remove the "*")
  ---------------------------------------------------;

       data out.results;
         set &result ;
       run;

/* end of BOOTVARE_V31.SAS program  */
```

Results:

The following tables present the results of the analyses done using the BOOTVAR.SAS program as defined in the example. Please refer to the fact sheets of Appendix A for a more complete description of the results.

Variance Estimation for a TOTAL
using 500 bootstrap replicates

prc8_cur	Variable	Sample size	Total	Standard error	Coeff. of variation	Lower limit Confidence interval 95%	Upper limit confidence interval 95%
All	mdiab	392	378528.24	20925.33	5.53	337515.35	419541.13
All	fdiab	393	320320.34	18493.92	5.77	284072.92	356567.76
10	mdiab	35	7029.13	1380.61	19.64	4323.19	9735.07
10	fdiab	64	13712.19	1604.37	11.70	10567.68	16856.70
24	mdiab	104	110452.80	10944.25	9.91	89002.46	131903.14
24	fdiab	95	94839.47	11271.72	11.89	72747.31	116931.63
35	mdiab	190	198237.67	15854.33	8.00	167163.75	229311.59
35	fdiab	184	164201.93	13524.96	8.24	137693.50	190710.36
59	mdiab	63	62808.64	8568.12	13.64	46015.43	79601.85
59	fdiab	50	47566.75	7277.83	15.30	33302.46	61831.04

Variance Estimation for a RATIO
using 500 bootstrap replicates

prc8_cur	Numerator	Denominator	Numerator size	Ratio	Standard error	Coeff. of variation	Lower limit Confidence interval 95%	Upper limit Confidence interval 95%
All	mdiab	males	392	0.0335	0.0019	5.53	0.0299	0.0371
All	fdiab	females	393	0.0277	0.0016	5.77	0.0246	0.0309
10	mdiab	males	35	0.0263	0.0052	19.64	0.0162	0.0365
10	fdiab	females	64	0.0506	0.0059	11.70	0.0390	0.0622
24	mdiab	males	104	0.0312	0.0031	9.91	0.0252	0.0373
24	fdiab	females	95	0.0262	0.0031	11.89	0.0201	0.0323
35	mdiab	males	190	0.0357	0.0029	8.00	0.0301	0.0413
35	fdiab	females	184	0.0288	0.0024	8.24	0.0241	0.0334
59	mdiab	males	63	0.0324	0.0044	13.64	0.0237	0.0411
59	fdiab	females	50	0.0242	0.0037	15.30	0.0170	0.0315

Variance Estimation for a DIFFERENCE BETWEEN RATIOS
using 500 bootstrap replicates

prc8_cur	Num1	Den1	Num2	Den2	Num1 size	Num2 size	Difference of ratios	z	p value	Std. err.	C.V.	Lower limit confidence interval 95%	Upper limit confidence interval 95%
All	mdiab	males	fdiab	females	392	393	0.0058	2.31	0.0210	0.0025	43.34	0.0009	0.0107
10	mdiab	males	fdiab	females	35	64	-0.0243	-2.77	0.0055	0.0088	36.05	-0.0414	-0.0071
24	mdiab	males	fdiab	females	104	95	0.0050	1.16	0.2440	0.0043	85.84	-0.0034	0.0135
35	mdiab	males	fdiab	females	190	184	0.0069	1.77	0.0775	0.0039	56.64	-0.0008	0.0145
59	mdiab	males	fdiab	females	63	50	0.0082	1.37	0.1716	0.0060	73.15	-0.0035	0.0198

Variance Estimation for a PERCENTILE using 500 bootstrap replicates

prc8_cur	Variable	Sample size	Percentile (1-99)	Percentile value	Standard error	Coeff. of variation	Lower limit confidence interval 95%	Upper limit confidence interval 95%
10	dhc8_age	2844	75	50.00	0.46	0.91	49.10	50.90
24	dhc8_age	8221	75	51.00	0.50	0.98	50.02	51.98
35	dhc8_age	13334	75	51.00	0.09	0.17	50.83	51.17
59	dhc8_age	4602	75	52.00	0.51	0.98	51.00	53.00

Variance Estimation for a LINEAR REGRESSION using 500 bootstrap replicates

----------------------------------- Model=1: Dependent Variable = dhc8_age ----------------------------------

prc8_cur	Independent variables	Beta	Standard error	t	p value
10	Intercept	39.6002	0.1939	204.21	0.0000
10	diab	17.8948	1.9821	9.03	0.0000
10	females	0.5451	0.2936	1.86	0.0634
24	Intercept	40.5369	0.1687	240.24	0.0000
24	diab	18.6715	1.0860	17.19	0.0000
24	females	1.9133	0.1896	10.09	0.0000
35	Intercept	40.2896	0.1159	347.69	0.0000
35	diab	19.6350	0.9179	21.39	0.0000
35	females	1.7693	0.1472	12.02	0.0000
59	Intercept	40.8565	0.2043	199.99	0.0000
59	diab	19.2997	1.5238	12.67	0.0000
59	females	1.5162	0.2189	6.93	0.0000

Variance Estimation for a LOGISTIC REGRESSION using 500 bootstrap replicates

------------------------------------ Model=1: Dependent Variable = diab ------------------------------------

prc8_cur	Independent variables	Beta	Odds ratio	Standard error	Wald	p value	Odds ratio lower limit conf. int. 95%	Odds ratio upper limit conf. int. 95%
10	Intercept	-5.9903	0.00	0.4321	192.16	0.0000	0.0011	0.0058
10	females	0.6332	1.88	0.2830	5.01	0.0252	1.0818	3.2802
10	DHC8_AGE	0.0525	1.05	0.0064	67.17	0.0000	1.0408	1.0672
24	Intercept	-6.0119	0.00	0.2334	663.39	0.0000	0.0016	0.0039
24	females	-0.3221	0.72	0.1655	3.79	0.0517	0.5239	1.0024
24	DHC8_AGE	0.0553	1.06	0.0036	231.04	0.0000	1.0494	1.0645
35	Intercept	-5.9343	0.00	0.2038	847.62	0.0000	0.0018	0.0039
35	females	-0.3885	0.68	0.1332	8.51	0.0035	0.5223	0.8803
35	DHC8_AGE	0.0568	1.06	0.0032	321.10	0.0000	1.0518	1.0650
59	Intercept	-5.9866	0.00	0.3175	355.62	0.0000	0.0013	0.0047
59	females	-0.4421	0.64	0.2385	3.44	0.0637	0.4027	1.0256
59	DHC8_AGE	0.0550	1.06	0.0049	124.78	0.0000	1.0464	1.0668

Rao-Scott Second order adjusted CHI-SQUARE test of indepedence using 500 bootstrap replicates

---------------------------------- Test=1:   Variables: diab VS females ----------------------------------

prc8_cur	Statistic	Deg. of freedom	Chi-square value	p value
10	R.-S. sec. order adj. chi-2	1	6.41	0.0113
24	R.-S. sec. order adj. chi-2	1	1.29	0.2564
35	R.-S. sec. order adj. chi-2	1	2.97	0.0849
59	R.-S. sec. order adj. chi-2	1	1.70	0.1919

Note: An asterisk (*) in the column Statistic indicates that the table has cell(s) with less than 5 unweighted observations. Chi-square may not be a valid test in that case.

CHI-SQUARE test of indepedence: Various statistics for the two-way table using 500 bootstrap replicates

---------------------------------- Test=1:   Variables: diab VS females ----------------------------------

prc8_cur	diab	Statistic	females_0	females_1	total
10	.	Pourcentage	0.4938	0.5062	.
10	.	Variance	0.0000	0.0000	.
10	.	_C.V._	0.0000	0.0000	.
10	0	Pourcentage	0.4787	0.4765	0.9552
10	0	Variance	0.0000	0.0000	0.0000
10	0	_C.V._	0.0062	0.0073	0.0042
10	1	Pourcentage	0.0152	0.0296	0.0448
10	1	Variance	0.0000	0.0000	0.0000
10	1	_C.V._	0.1965	0.1170	0.0890
24	.	Pourcentage	0.4915	0.5085	.
24	.	Variance	0.0000	0.0000	.
24	.	_C.V._	0.0009	0.0008	.
24	0	Pourcentage	0.4734	0.4929	0.9663
24	0	Variance	0.0000	0.0000	0.0000
24	0	_C.V._	0.0039	0.0039	0.0027
24	1	Pourcentage	0.0182	0.0156	0.0337
24	1	Variance	0.0000	0.0000	0.0000
24	1	_C.V._	0.0998	0.1188	0.0779
35	.	Pourcentage	0.4899	0.5101	.
35	.	Variance	0.0000	0.0000	.
35	.	_C.V._	0.0001	0.0001	.
35	0	Pourcentage	0.4690	0.4927	0.9617
35	0	Variance	0.0000	0.0000	0.0000
35	0	_C.V._	0.0036	0.0029	0.0022
35	1	Pourcentage	0.0210	0.0174	0.0383
35	1	Variance	0.0000	0.0000	0.0000
35	1	_C.V._	0.0800	0.0824	0.0544
59	.	Pourcentage	0.4943	0.5057	.
59	.	Variance	0.0000	0.0000	.
59	.	_C.V._	0.0003	0.0003	.
59	0	Pourcentage	0.4754	0.4914	0.9668
59	0	Variance	0.0000	0.0000	0.0000
59	0	_C.V._	0.0054	0.0044	0.0034
59	1	Pourcentage	0.0189	0.0143	0.0332
59	1	Variance	0.0000	0.0000	0.0000
59	1	_C.V._	0.1365	0.1532	0.0983